*==============================================================================*
*===*   Title: Framing Fact-Checks as a "confirmation" increases engagement
*===*          with corrections of misinformation: a four-country study    
*===*
*===* Authors: Natalia Aruguete, Flavia Batista, Ernesto Calvo,            
*===*          Matias Guizzo Altube, Carlos Scartascini, and Tiago Ventura 
*===*
*===*    Date: January 2024                                                
*===*
*===* Journal: Scientific Reports                                          
*==============================================================================*

* This do-file contains the replication code for the analysis conducted in the
* above-mentioned paper. The code generates all the tables and figures included
* in the paper as well as in the Supplementary Information File (SIF).



****===========================================================================*
**#*===* Index
*===* 0. Preamble
*===* 1. Tables
*===*   1.1. Tables in main article
*===*       1.1.1. Descriptive statistics by country (T1)
*===*       1.1.2. Difference of means between confirmation and refutation frames (T2)
*===*   1.2. Tables in SIF
*===*       1.2.1. Balance tests for covariables (S1-S4)
*===*       1.2.2. Differences of means in the general specification by country (S5-S8)
*===*       1.2.3. Differences of means in secondary specifications (S9-S14)
*===*       1.2.4. Heterogeneity in reactions and emotions (S15-S16)
*===*       1.2.5. Detailed heterogeneity by partisan attachment (S17-S18)
*===*       1.2.6. Detailed heterogeneity by vaccination status (S179-S20)
*===*       1.2.7. Detailed heterogeneity by education (S21-S22)
*===*       1.2.8. Detailed heterogeneity by time to read (S23-S24)
*===*       1.2.9. Detailed heterogeneity by attention filter (S25-S26)
*===*       1.2.10. Placebo experiment by country (S29-S31)
*===* 2. Figures
*===*   2.1. Figures in main article
*===*       2.1.1. Framing effects on reactions and emotions by country
*===*       2.1.2. Engagement rate by vaccine and frame in Argentina
*===*       2.1.3. Engagement rate by use of labels in Brazil, Chile, and Colombia
*===*       2.1.4. Engagement rate in placebo in Brazil, Chile, and Colombia
*===*       2.1.5. Probability of engagement by time to read
*===*   2.2. Figures in SIF



****===========================================================================*
**#*===* 0. Preamble

// error // Uncomment this line to prevent the code from accidentally running when editing

global main = "~\Framing Fact-Checks" // Write the path to the main folder here
	global figures = "$main\Figures"
	global tables = "$main\Tables"

* The current date will be used to create specific folders for output
global today = strtrim("`=c(current_date)'")

* Globals with options for figures
global gph_gral_op "graphreg(col(white)) ylab(, ang(h) tlen(0) glc(gray*.1)) ysc(lcol(none)) xlab(, tlen(0)) ti(, col(black) pos(11)) subti(, pos(11)) legend(region(lcol(white)) size(small) pos(12) symxsize(*.3) symysize(*.6))"
global gph_by_op " by(, graphreg(col(white)) ti(, col(black) pos(11))) subti(, bcol(gray*.1) pos(11)) ysc(lcol(none)) ylab(, ang(h) tlen(0) glc(gray*.1)) xlab(, tlen(0)) legend(region(lcol(white)) size(small) pos(12) symxsize(*.3) symysize(*.6))"

* Open dataset
clear all
use "$main/FramingFactChecks_Data.dta"



****===========================================================================*
**#*===* 1. Tables

cap n mkdir "$tables/$today"
	cd "$tables/$today"


**#*===*   1.1. Tables in main article

**#*===*      1.1.1. Descriptive Statistics by Country (T1)

* Number of observations in each country
tab country if filter == 1
// ARG 2,349
// BRA 2,401
// CHL 2,384
// COL 2,378

preserve
* Binary variables multiplied by 100 for showing means as percentages in Table 1
foreach v of varlist educg1-educg5 woman works government opposition covid vaxsh1-vaxsh3 {
	replace `v' = 100*`v'
}

mat T1 = J(15, 8, .)

local controls = "age educg1-educg5 woman works government opposition ln_time_read covid vaxsh1-vaxsh3" // income pol_know

	local j = 1
	foreach cn in "ARG" "BRA" "CHL" "COL" {
		
		local i = 1
		foreach v of varlist `controls' {
			qui: sum `v' if country == "`cn'" & filter == 1
				mat T1[`i', `j'] = r(mean)
				mat T1[`i', `j'+1] = r(sd)

			local i = `i' + 1
		}
		
		local j = `j' + 2
	}

frmttable using "Table1", tex frag replace ///
	statmat(T1) sdec(2) ///
	cti("" "Argentina" "" "Brazil" "" "Chile" "" "Colombia" \ ///
		"Variable" "(N = 2,349)" "" "(N = 2,401)" "" "(N = 2,384)" "" "(N = 2,378)") ///
	rti("Age (years)" \ "\hspace{1ex} Incomplete secondary (or -)" \ "\hspace{1ex} Completed secondary" \ "\hspace{1ex} Incomplete college" \ "\hspace{1ex} Completed college" \ "\hspace{1ex} Incomplete graduate (or +)" \ "Woman" \ "Employed" \ ///
		 "Vote for incumbent party" \ "Vote for opposition party" \ "Time to read (log of seconds)" \ "Have had COVID-19" \ "\hspace{1ex} Non-vaccinated against COVID-19" \ "\hspace{1ex} Vaccinated once" \ "\hspace{1ex} Vaccinated twice (or +)") ///
	noblank ///
	hline(101 {0} 1) ///
	note("\smallskip\textit{Note}: Coefficients represent the average value of each control variable in the complete sample from each country survey. Standard deviations in parentheses.")
restore


**#*===*      1.1.2. Difference of Means between Confirmation and Refutation Frames (T2)

mat T2 = J(14, 8, .)
mat T2star = J(14, 8, 0)

local controls = "woman i.age i.education works i.vote_encode ln_time_read i.vax_shots covid" // income pol_know
local depvars = "engage like share comment optimism joy anger sadness stress fear disgust indifference"
	
	local j = 1
	foreach cn in "ARG" "BRA" "CHL" "COL" {
		
		local i = 2
		foreach v in `depvars' {
			qui: reg `v' true `controls' if country == "`cn'" & treatment != "Dog" & filter == 1, robust
				mat results = r(table)
					mat T2[`i', `j'] = results["b","true"]
					mat T2[`i', `j'+1] = results["se","true"]
				local pval = results["pvalue","true"]
					mat T2star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)

			if (`i' == 5) local i = `i' + 2
			else local i = `i' + 1
		}
		
		local j = `j' + 2
	}


frmttable using "Table2", tex frag replace ///
	statmat(T2) substat(1) sdec(3) ///
	annotate(T2star) asym("*", "**", "***") ///
	cti("Variable" "Argentina" "Brazil" "Chile" "Colombia") ///
	rti("\textit{Reactions}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
		"\noalign{\smallskip}\textit{Emotions}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ "\hspace{1ex} Stressed" \ "" \ ///
		"\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "") ///
	noblank ///
	hline(11 0 0 0 0 0 0 0 0 0 1 {0} 1) ///
	note("\smallskip\textit{Note}: Robust standard errors in parentheses."  "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Full set of models in the SIF file to this article.")


	
**#*===*   1.2. Tables in SIF

**#*===*      1.2.1. Balance tests for covariables (S1-S4)

* Table S1. Balance tests in the general specification by country

preserve
* Binary variables multiplied by 100 for showing means as percentages in Table 1
foreach v of varlist educg1-educg5 woman works government opposition covid vaxsh1-vaxsh3 {
	replace `v' = 100*`v'
}

mat TS1 = J(64, 6, .)
mat TS1star = J(64, 6, 0)

local controls = "age educg1-educg5 woman works government opposition ln_time_read covid vaxsh1-vaxsh3"
	
	local i = 2
	foreach cn in ARG BRA CHL COL {
		foreach v of varlist `controls' {
			* Mean values
			qui: ttest `v' if country == "`cn'" & treatment != "Dog" & filter == 1, by(true)
				mat TS1[`i', 1] = r(mu_1)
				mat TS1[`i', 2] = (r(sd_1)/sqrt(r(N_1)))
				mat TS1[`i', 3] = r(mu_2)
				mat TS1[`i', 4] = r(sd_2)/sqrt(r(N_2))
			
			* Difference
			qui: reg `v' true if country == "`cn'" & treatment != "Dog" & filter == 1, robust
				mat results = r(table)
					mat TS1[`i', 5] = results["b","true"]
					mat TS1[`i', 6] = results["se","true"]
				local pval = results["pvalue","true"]
					mat TS1star[`i', 5] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)

			if (mod(`i',16) == 0) local i = `i' + 2
			else local i = `i' + 1
		}
	}

frmttable using "TableS1", tex frag replace ///
	statmat(TS1) sdec(2) ///
	annotate(TS1star) asym("*", "**", "***") ///
	cti("" "Refutation" "" "Confirmation" "" "" "" \ ///
		"Variable" "frame" "" "frame" "" "Difference" "") ///
	rti("\textit{Argentina}" \ "Age (years)" \ "\hspace{1ex} Incomplete secondary (or -)" \ "\hspace{1ex} Completed secondary" \ "\hspace{1ex} Incomplete college" \ "\hspace{1ex} Completed college" \ "\hspace{1ex} Incomplete graduate (or +)" \ "Woman" \ "Employed" \ ///
		 "Vote for incumbent party" \ "Vote for opposition party" \ "Time to read (log of seconds)" \ "Have had COVID-19" \ "\hspace{1ex} Non-vaccinated against COVID-19" \ "\hspace{1ex} Vaccinated once" \ "\hspace{1ex} Vaccinated twice (or +)" \ ///
		"\noalign{\smallskip}\textit{Brazil}" \ "Age (years)" \ "\hspace{1ex} Incomplete secondary (or -)" \ "\hspace{1ex} Completed secondary" \ "\hspace{1ex} Incomplete college" \ "\hspace{1ex} Completed college" \ "\hspace{1ex} Incomplete graduate (or +)" \ "Woman" \ "Employed" \ ///
		 "Vote for incumbent party" \ "Vote for opposition party" \ "Time to read (log of seconds)" \ "Have had COVID-19" \ "\hspace{1ex} Non-vaccinated against COVID-19" \ "\hspace{1ex} Vaccinated once" \ "\hspace{1ex} Vaccinated twice (or +)" \ ///
		"\noalign{\smallskip}\textit{Chile}" \ "Age (years)" \ "\hspace{1ex} Incomplete secondary (or -)" \ "\hspace{1ex} Completed secondary" \ "\hspace{1ex} Incomplete college" \ "\hspace{1ex} Completed college" \ "\hspace{1ex} Incomplete graduate (or +)" \ "Woman" \ "Employed" \ ///
		 "Vote for incumbent party" \ "Vote for opposition party" \ "Time to read (log of seconds)" \ "Have had COVID-19" \ "\hspace{1ex} Non-vaccinated against COVID-19" \ "\hspace{1ex} Vaccinated once" \ "\hspace{1ex} Vaccinated twice (or +)" \ ///
		"\noalign{\smallskip}\textit{Colombia}" \ "Age (years)" \ "\hspace{1ex} Incomplete secondary (or -)" \ "\hspace{1ex} Completed secondary" \ "\hspace{1ex} Incomplete college" \ "\hspace{1ex} Completed college" \ "\hspace{1ex} Incomplete graduate (or +)" \ "Woman" \ "Employed" \ ///
		 "Vote for incumbent party" \ "Vote for opposition party" \ "Time to read (log of seconds)" \ "Have had COVID-19" \ "\hspace{1ex} Non-vaccinated against COVID-19" \ "\hspace{1ex} Vaccinated once" \ "\hspace{1ex} Vaccinated twice (or +)") ///
	noblank ///
	hline(101 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 {0} 1) ///
	note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Significance levels of differences are derived from linear regression models.")
restore


* Table S2. Balance test by country and use of labels

preserve
* Binary variables multiplied by 100 for showing means as percentages in Table 1
foreach v of varlist educg1-educg5 woman works government opposition covid vaxsh1-vaxsh3 {
	replace `v' = 100*`v'
}

mat TS2 = J(48, 12, .)
mat TS2star = J(48, 12, 0)

local controls = "age educg1-educg5 woman works government opposition ln_time_read covid vaxsh1-vaxsh3"
	
	local i = 2
	foreach cn in BRA CHL COL {
		foreach v of varlist `controls' {
			
			*** With labels
			* Mean values
			qui: ttest `v' if country == "`cn'" & treatment == "Label" & filter == 1, by(true)
				mat TS2[`i', 1] = r(mu_1)
				mat TS2[`i', 2] = (r(sd_1)/sqrt(r(N_1)))
				mat TS2[`i', 3] = r(mu_2)
				mat TS2[`i', 4] = r(sd_2)/sqrt(r(N_2))
			
			* Difference
			qui: reg `v' true if country == "`cn'" & treatment == "Label" & filter == 1 , robust
				mat results = r(table)
					mat TS2[`i', 5] = results["b","true"]
					mat TS2[`i', 6] = results["se","true"]
				local pval = results["pvalue","true"]
					mat TS2star[`i', 5] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
					
			*** Without labels
			* Mean values
			qui: ttest `v' if country == "`cn'" & treatment == "No-label" & filter == 1, by(true)
				mat TS2[`i', 7] = r(mu_1)
				mat TS2[`i', 8] = (r(sd_1)/sqrt(r(N_1)))
				mat TS2[`i', 9] = r(mu_2)
				mat TS2[`i', 10] = r(sd_2)/sqrt(r(N_2))
			
			* Difference
			qui: reg `v' true if country == "`cn'" & treatment == "No-label" & filter == 1, robust
				mat results = r(table)
					mat TS2[`i', 11] = results["b","true"]
					mat TS2[`i', 12] = results["se","true"]
				local pval = results["pvalue","true"]
					mat TS2star[`i', 11] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)

			if (mod(`i',16) == 0) local i = `i' + 2
			else local i = `i' + 1
		}
	}

frmttable using "TableS2", tex frag replace ///
	statmat(TS2) sdec(2) ///
	annotate(TS2star) asym("*", "**", "***") ///
	cti("" "With labels" "" "" "" "" "" "Without labels" "" "" "" "" "" \ ///
		"Variable" "Refutation" "" "Confirmation" "" "Difference" "" "Refutation" "" "Confirmation" "" "Difference" "") ///
	rti("\textit{Brazil}" \ "Age (years)" \ "\hspace{1ex} Incomplete secondary (or -)" \ "\hspace{1ex} Completed secondary" \ "\hspace{1ex} Incomplete college" \ "\hspace{1ex} Completed college" \ "\hspace{1ex} Incomplete graduate (or +)" \ "Woman" \ "Employed" \ ///
		 "Vote for incumbent party" \ "Vote for opposition party" \ "Time to read (log of seconds)" \ "Have had COVID-19" \ "\hspace{1ex} Non-vaccinated against COVID-19" \ "\hspace{1ex} Vaccinated once" \ "\hspace{1ex} Vaccinated twice (or +)" \ ///
		"\noalign{\smallskip}\textit{Chile}" \ "Age (years)" \ "\hspace{1ex} Incomplete secondary (or -)" \ "\hspace{1ex} Completed secondary" \ "\hspace{1ex} Incomplete college" \ "\hspace{1ex} Completed college" \ "\hspace{1ex} Incomplete graduate (or +)" \ "Woman" \ "Employed" \ ///
		 "Vote for incumbent party" \ "Vote for opposition party" \ "Time to read (log of seconds)" \ "Have had COVID-19" \ "\hspace{1ex} Non-vaccinated against COVID-19" \ "\hspace{1ex} Vaccinated once" \ "\hspace{1ex} Vaccinated twice (or +)" \ ///
		"\noalign{\smallskip}\textit{Colombia}" \ "Age (years)" \ "\hspace{1ex} Incomplete secondary (or -)" \ "\hspace{1ex} Completed secondary" \ "\hspace{1ex} Incomplete college" \ "\hspace{1ex} Completed college" \ "\hspace{1ex} Incomplete graduate (or +)" \ "Woman" \ "Employed" \ ///
		 "Vote for incumbent party" \ "Vote for opposition party" \ "Time to read (log of seconds)" \ "Have had COVID-19" \ "\hspace{1ex} Non-vaccinated against COVID-19" \ "\hspace{1ex} Vaccinated once" \ "\hspace{1ex} Vaccinated twice (or +)") ///
	noblank ///
	hline(101 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 {0} 1) ///
	note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Significance levels of differences are derived from linear regression models.")
restore


* Table S3. Balance test by mentioned vaccine brand in Argentina

preserve
* Binary variables multiplied by 100 for showing means as percentages in Table 1
foreach v of varlist educg1-educg5 woman works government opposition covid vaxsh1-vaxsh3 {
	replace `v' = 100*`v'
}

mat TS3 = J(48, 6, .)
mat TS3star = J(48, 6, 0)

local controls = "age educg1-educg5 woman works government opposition ln_time_read covid vaxsh1-vaxsh3" // income pol_know
	
	local i = 2
	foreach vax in Moderna Sputnik AstraZeneca {
		foreach v of varlist `controls' {
			* Mean values
			qui: ttest `v' if country == "ARG" & treatment == "`vax'" & filter == 1, by(true)
				mat TS3[`i', 1] = r(mu_1)
				mat TS3[`i', 2] = (r(sd_1)/sqrt(r(N_1)))
				mat TS3[`i', 3] = r(mu_2)
				mat TS3[`i', 4] = r(sd_2)/sqrt(r(N_2))
			
			* Difference
			qui: reg `v' true if country == "ARG" & treatment == "`vax'" & filter == 1, robust
				mat results = r(table)
					mat TS3[`i', 5] = results["b","true"]
					mat TS3[`i', 6] = results["se","true"]
				local pval = results["pvalue","true"]
					mat TS3star[`i', 5] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)

			if (mod(`i',16) == 0) local i = `i' + 2
			else local i = `i' + 1
		}
	}

frmttable using "TableS3", tex frag replace ///
	statmat(TS3) sdec(2) ///
	annotate(TS3star) asym("*", "**", "***") ///
	cti("" "Refutation" "" "Confirmation" "" "" "" \ ///
		"Variable" "frame" "" "frame" "" "Difference" "") ///
	rti("\textit{Moderna}" \ "Age (years)" \ "\hspace{1ex} Incomplete secondary (or -)" \ "\hspace{1ex} Completed secondary" \ "\hspace{1ex} Incomplete college" \ "\hspace{1ex} Completed college" \ "\hspace{1ex} Incomplete graduate (or +)" \ "Woman" \ "Employed" \ ///
		 "Vote for incumbent party" \ "Vote for opposition party" \ "Time to read (log of seconds)" \ "Have had COVID-19" \ "\hspace{1ex} Non-vaccinated against COVID-19" \ "\hspace{1ex} Vaccinated once" \ "\hspace{1ex} Vaccinated twice (or +)" \ ///
		"\noalign{\smallskip}\textit{Sputnik}" \ "Age (years)" \ "\hspace{1ex} Incomplete secondary (or -)" \ "\hspace{1ex} Completed secondary" \ "\hspace{1ex} Incomplete college" \ "\hspace{1ex} Completed college" \ "\hspace{1ex} Incomplete graduate (or +)" \ "Woman" \ "Employed" \ ///
		 "Vote for incumbent party" \ "Vote for opposition party" \ "Time to read (log of seconds)" \ "Have had COVID-19" \ "\hspace{1ex} Non-vaccinated against COVID-19" \ "\hspace{1ex} Vaccinated once" \ "\hspace{1ex} Vaccinated twice (or +)" \ ///
		"\noalign{\smallskip}\textit{AstraZeneca}" \ "Age (years)" \ "\hspace{1ex} Incomplete secondary (or -)" \ "\hspace{1ex} Completed secondary" \ "\hspace{1ex} Incomplete college" \ "\hspace{1ex} Completed college" \ "\hspace{1ex} Incomplete graduate (or +)" \ "Woman" \ "Employed" \ ///
		 "Vote for incumbent party" \ "Vote for opposition party" \ "Time to read (log of seconds)" \ "Have had COVID-19" \ "\hspace{1ex} Non-vaccinated against COVID-19" \ "\hspace{1ex} Vaccinated once" \ "\hspace{1ex} Vaccinated twice (or +)") ///
	noblank ///
	hline(101 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 {0} 1) ///
	note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Significance levels of differences are derived from linear regression models.")
restore


* Table S4. Balance test in the dog placebo experiment by country

preserve
* Binary variables multiplied by 100 for showing means as percentages in Table 1
foreach v of varlist educg1-educg5 woman works government opposition covid vaxsh1-vaxsh3 {
	replace `v' = 100*`v'
}

mat TS4 = J(48, 6, .)
mat TS4star = J(48, 6, 0)

local controls = "age educg1-educg5 woman works government opposition ln_time_read covid vaxsh1-vaxsh3" // income pol_know
	
	local i = 2
	foreach cn in BRA CHL COL {
		foreach v of varlist `controls' {
			* Mean values
			qui: ttest `v' if country == "`cn'" & treatment == "Dog" & filter == 1, by(true)
				mat TS4[`i', 1] = r(mu_1)
				mat TS4[`i', 2] = (r(sd_1)/sqrt(r(N_1)))
				mat TS4[`i', 3] = r(mu_2)
				mat TS4[`i', 4] = r(sd_2)/sqrt(r(N_2))
			
			* Difference
			qui: reg `v' true if country == "`cn'" & treatment == "Dog" & filter == 1, robust
				mat results = r(table)
					mat TS4[`i', 5] = results["b","true"]
					mat TS4[`i', 6] = results["se","true"]
				local pval = results["pvalue","true"]
					mat TS4star[`i', 5] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)

			if (mod(`i',16) == 0) local i = `i' + 2
			else local i = `i' + 1
		}
	}

frmttable using "TableS4", tex frag replace ///
	statmat(TS4) sdec(2) ///
	annotate(TS4star) asym("*", "**", "***") ///
	cti("" "Refutation" "Confirmation" "" \ ///
		"Variable" "frame" "frame" "Difference") ///
	rti("\textit{Brazil}" \ "Age (years)" \ "\hspace{1ex} Incomplete secondary (or -)" \ "\hspace{1ex} Completed secondary" \ "\hspace{1ex} Incomplete college" \ "\hspace{1ex} Completed college" \ "\hspace{1ex} Incomplete graduate (or +)" \ "Woman" \ "Employed" \ ///
		 "Vote for incumbent party" \ "Vote for opposition party" \ "Time to read (log of seconds)" \ "Have had COVID-19" \ "\hspace{1ex} Non-vaccinated against COVID-19" \ "\hspace{1ex} Vaccinated once" \ "\hspace{1ex} Vaccinated twice (or +)" \ ///
		"\noalign{\smallskip}\textit{Chile}" \ "Age (years)" \ "\hspace{1ex} Incomplete secondary (or -)" \ "\hspace{1ex} Completed secondary" \ "\hspace{1ex} Incomplete college" \ "\hspace{1ex} Completed college" \ "\hspace{1ex} Incomplete graduate (or +)" \ "Woman" \ "Employed" \ ///
		 "Vote for incumbent party" \ "Vote for opposition party" \ "Time to read (log of seconds)" \ "Have had COVID-19" \ "\hspace{1ex} Non-vaccinated against COVID-19" \ "\hspace{1ex} Vaccinated once" \ "\hspace{1ex} Vaccinated twice (or +)" \ ///
		"\noalign{\smallskip}\textit{Colombia}" \ "Age (years)" \ "\hspace{1ex} Incomplete secondary (or -)" \ "\hspace{1ex} Completed secondary" \ "\hspace{1ex} Incomplete college" \ "\hspace{1ex} Completed college" \ "\hspace{1ex} Incomplete graduate (or +)" \ "Woman" \ "Employed" \ ///
		 "Vote for incumbent party" \ "Vote for opposition party" \ "Time to read (log of seconds)" \ "Have had COVID-19" \ "\hspace{1ex} Non-vaccinated against COVID-19" \ "\hspace{1ex} Vaccinated once" \ "\hspace{1ex} Vaccinated twice (or +)") ///
	noblank ///
	hline(101 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 {0} 1) ///
	note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Significance levels of differences are derived from linear regression models.")
restore



**#*===*      1.2.2. Differences of means in the general specification by country (S5-S8)

local controls = "woman i.age i.education works i.vote_encode ln_time_read covid i.vax_shots" // income pol_know
local depvars = "engage like share comment optimism joy anger sadness stress fear disgust indifference"

local k = 5
foreach cn in "ARG" "BRA" "CHL" "COL" {
	mat TS`k' = J(14, 8, .)
	mat TS`k'star = J(14, 8, 0)

	local i = 2
	foreach v in `depvars' {

		* Mean values
		qui: ttest `v' if country == "`cn'" & treatment != "Dog" & filter == 1, by(true)
			mat TS`k'[`i', 1] = r(mu_1)
			mat TS`k'[`i', 2] = (r(sd_1)/sqrt(r(N_1)))
			mat TS`k'[`i', 3] = r(mu_2)
			mat TS`k'[`i', 4] = r(sd_2)/sqrt(r(N_2))
		
		* Difference without controls
		qui: reg `v' true if country == "`cn'" & treatment != "Dog" & filter == 1, robust
			mat results = r(table)
				mat TS`k'[`i', 5] = results["b","true"]
				mat TS`k'[`i', 6] = results["se","true"]
			local pval = results["pvalue","true"]
				mat TS`k'star[`i', 5] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)

		* Demographic and political controls + time to read
		qui: reg `v' true `controls' if country == "`cn'" & treatment != "Dog" & filter == 1, robust
			mat results = r(table)
				mat TS`k'[`i',7] = results["b","true"]
				mat TS`k'[`i',8] = results["se","true"]
			local pval = results["pvalue","true"]
				mat TS`k'star[`i',7] = (`pval' < .05) + (`pval' < .01) + (`pval' < .001)

		if (`i' == 5) local i = `i' + 2
		else local i = `i' + 1
	}

	frmttable using "TableS`k'", tex frag replace ///
		statmat(TS`k') substat(1) sdec(3) ///
		annotate(TS`k'star) asym("*", "**", "***") ///
		cti("" "Refutation" "Confirmation" "Simple" "Difference" \ ///
			"Variable" "frame" "frame" "difference" "with controls") ///
		rti("\textit{Reactions}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Emotions}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent") ///
		noblank ///
		hline(101 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Significance levels are derived from linear regression models." "The second difference is estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")

	local k = `k' + 1
}



**#*===*      1.2.3. Differences of means in secondary specifications (S9-S14)

***** Argentina (S9-S11)

* Table S9. Difference of means by mentioned vaccine brand

mat TS9 = J(14, 8, .)
mat TS9star = J(14, 8, 0)

local controls = "woman i.age i.education works i.vote_encode ln_time_read covid i.vax_shots" // income pol_know
local depvars = "engage like share comment optimism joy anger sadness stress fear disgust indifference"

	local j = 1
	foreach vax in "Overall" "Moderna" "Sputnik" "AstraZeneca" {
		
		local i = 2
		foreach v in `depvars' {

			if (`j' == 1) qui: reg `v' true `controls' if country == "ARG" & filter == 1, robust
			else qui: reg `v' true `controls' if country == "ARG" & treatment == "`vax'" & filter == 1, robust
				mat results = r(table)
					mat TS9[`i',`j'] = results["b","true"]
					mat TS9[`i',`j'+1] = results["se","true"]
				local pval = results["pvalue","true"]
					mat TS9star[`i',`j'] = (`pval' < .05) + (`pval' < .01) + (`pval' < .001)

			if (`i' == 5) local i = `i' + 2
			else local i = `i' + 1
		}
		local j = `j' + 2
	}
	
frmttable using "TableS9", tex frag replace ///
		statmat(TS9) substat(1) sdec(3) ///
		annotate(TS9star) asym("*", "**", "***") ///
		cti("Variable" "Overall" "Moderna" "Sputnik" "AstraZeneca") ///
		rti("\textit{Reactions}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Emotions}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent") ///
		noblank ///
		hline(11 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")


* Table S10. Difference of means in reactions by mentioned vaccine brand and partisan attachment

mat TS10 = J(15, 8, .)
mat TS10star = J(15, 8, 0)

local controls = "woman i.age i.education works i.vote_encode ln_time_read covid i.vax_shots" // income pol_know
local depvars = "engage like share comment"

	local j = 1
	foreach vax in "Overall" "Moderna" "Sputnik" "AstraZeneca" {
		
		local i = 2
		foreach condition in "government == 1" "opposition == 1" "government == 0 & opposition == 0" {
			foreach v in `depvars' {

				if (`j' == 1) qui: reg `v' true `controls' if country == "ARG" & `condition' & filter == 1, robust
				else qui: reg `v' true `controls' if country == "ARG" & `condition' & treatment == "`vax'" & filter == 1, robust
					mat results = r(table)
						mat TS10[`i',`j'] = results["b","true"]
						mat TS10[`i',`j'+1] = results["se","true"]
					local pval = results["pvalue","true"]
						mat TS10star[`i',`j'] = (`pval' < .05) + (`pval' < .01) + (`pval' < .001)

				if (mod(`i', 5) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		local j = `j' + 2
	}

frmttable using "TableS10", tex frag replace ///
		statmat(TS10) substat(1) sdec(3) ///
		annotate(TS10star) asym("*", "**", "***") ///
		cti("Variable" "Overall" "Moderna" "Sputnik" "AstraZeneca") ///
		rti("\noalign{\smallskip}\textit{Government (Frente de Todos)}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Opposition (Juntos por el Cambio)}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{None (blank vote)}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment") ///
		noblank ///
		hline(11 00 00 00 00 01 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")

		
* Table S11. Difference of means in emotions by mentioned vaccine brand and partisan attachment

mat TS11 = J(27, 8, .)
mat TS11star = J(27, 8, 0)

local controls = "woman i.age i.education works i.vote_encode ln_time_read covid i.vax_shots" // income pol_know
local depvars = "optimism joy anger sadness stress fear disgust indifference"

	local j = 1
	foreach vax in "Overall" "Moderna" "Sputnik" "AstraZeneca" {
		
		local i = 2
		foreach condition in "government == 1" "opposition == 1" "government == 0 & opposition == 0" {
			foreach v in `depvars' {

				if (`j' == 1) qui: reg `v' true `controls' if country == "ARG" & `condition' & filter == 1, robust
				else qui: reg `v' true `controls' if country == "ARG" & `condition' & treatment == "`vax'" & filter == 1, robust
					mat results = r(table)
						mat TS11[`i',`j'] = results["b","true"]
						mat TS11[`i',`j'+1] = results["se","true"]
					local pval = results["pvalue","true"]
						mat TS11star[`i',`j'] = (`pval' < .05) + (`pval' < .01) + (`pval' < .001)

				if (mod(`i', 9) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		local j = `j' + 2
	}

frmttable using "TableS11", tex frag replace ///
		statmat(TS11) substat(1) sdec(3) ///
		annotate(TS11star) asym("*", "**", "***") ///
		cti("Variable" "Overall" "Moderna" "Sputnik" "AstraZeneca") ///
		rti("\noalign{\smallskip}\textit{Government (Frente de Todos)}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Opposition (Juntos por el Cambio)}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{None (blank vote)}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent") ///
		noblank ///
		hline(11 00 00 00 00 00 00 00 00 01 00 00 00 00 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")


***** Brazil, Chile, and Colombia, by use of labels in vignette (S12-S14)

local controls = "woman i.age i.education works i.vote_encode ln_time_read covid i.vax_shots" // income pol_know
local depvars = "engage like share comment optimism joy anger sadness stress fear disgust indifference"

local k = 12
foreach cn in "BRA" "CHL" "COL" {
	mat TS`k' = J(14, 6, .)
	mat TS`k'star = J(14, 6, 0)

	local j = 1
	foreach tmt in "Overall" "Label" "No-label" {
		
		local i = 2
		foreach v in `depvars' {

			if (`j' == 1) qui: reg `v' true `controls' if country == "`cn'" & filter == 1, robust
			else qui: reg `v' true `controls' if country == "`cn'" & treatment == "`tmt'" & filter == 1, robust
				mat results = r(table)
					mat TS`k'[`i',`j'] = results["b","true"]
					mat TS`k'[`i',`j'+1] = results["se","true"]
				local pval = results["pvalue","true"]
					mat TS`k'star[`i',`j'] = (`pval' < .05) + (`pval' < .01) + (`pval' < .001)

			if (`i' == 5) local i = `i' + 2
			else local i = `i' + 1
		}
		local j = `j' + 2
	}
	
	frmttable using "TableS`k'", tex frag replace ///
		statmat(TS`k') substat(1) sdec(3) ///
		annotate(TS`k'star) asym("*", "**", "***") ///
		cti("Variable" "Overall" "With label" "Without label") ///
		rti("\textit{Reactions}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Emotions}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent") ///
		noblank ///
		hline(11 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")

	local k = `k' + 1
}



**#*===*      1.2.4. Heterogeneity in reactions and emotions (S15-S16)

* Heterogeneous differences of means in reactions
mat TS15 = J(20, 6, .)
mat TS15star = J(20, 6, 0)

local controls = "woman i.age college works i.vote_encode ln_time_read i.vax_shots covid"
local depvars = "engage like share comment"

	local j = 1
	foreach condition in "government == 1" "opposition == 1" "educ2 == 0" "educ2 == 1" "time_q == 1" "time_q == 5" {
		
		local i = 2
		foreach cn in "ARG" "BRA" "CHL" "COL" {
				
			foreach v in `depvars' {
				qui: reg `v' true `controls' if `condition' & country == "`cn'" & treatment != "Dog" & filter == 1, robust
					mat results = r(table)
						mat TS15[`i', `j'] = results["b","true"]
					local pval = results["pvalue","true"]
						mat TS15star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)

				if (mod(`i',5) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		
		local j = `j' + 1
	}

frmttable using "TableS15", tex frag replace ///
	statmat(TS15) sdec(3) ///
	annotate(TS15star) asym("*", "**", "***") ///
	cti("" "Partisan attachment" "" "Completed" "Incomplete" "Time to read" "" \ ///
		"Variable" "Government" "Opposition" "secondary (or -)" "college (or +)" "First quintile" "Fifth quintile") ///
	rti("\textit{Argentina}" \ "\hspace{1ex} Engage" \ "\hspace{1ex} Like" \ "\hspace{1ex} Share" \ "\hspace{1ex} Comment" \ ///
		"\noalign{\smallskip}\textit{Brazil}" \ "\hspace{1ex} Engage" \ "\hspace{1ex} Like" \ "\hspace{1ex} Share" \ "\hspace{1ex} Comment" \ ///
		"\noalign{\smallskip}\textit{Chile}" \ "\hspace{1ex} Engage" \ "\hspace{1ex} Like" \ "\hspace{1ex} Share" \ "\hspace{1ex} Comment" \ ///
		"\noalign{\smallskip}\textit{Colombia}" \ "\hspace{1ex} Engage" \ "\hspace{1ex} Like" \ "\hspace{1ex} Share" \ "\hspace{1ex} Comment") ///
	noblank ///
	hline(101 0 0 0 0 1 0 0 0 0 1 0 0 0 0 1 {0} 1) ///
	note("\smallskip\textit{Note}: Robust standard errors in parentheses."  "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Full set of models in the SIF file to this article.")


* Heterogeneous differences of means in emotions (S16)
mat TS16 = J(36, 6, .)
mat TS16star = J(36, 6, 0)

local controls = "woman i.age i.education works i.vote_encode ln_time_read covid i.vax_shots" // income pol_know
local depvars = "optimism joy anger sadness stress fear disgust indifference"

	local j = 1
	foreach condition in "government == 1" "opposition == 1" "educ2 == 0" "educ2 == 1" "time_q == 1" "time_q == 5" {
		
		local i = 2
		foreach cn in "ARG" "BRA" "CHL" "COL" {
				
			foreach v in `depvars' {
				qui: reg `v' true `controls' if `condition' & country == "`cn'" & treatment != "Dog" & filter == 1, robust
					mat results = r(table)
						mat TS16[`i', `j'] = results["b","true"]
					local pval = results["pvalue","true"]
						mat TS16star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)

				if (mod(`i',9) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		
		local j = `j' + 1
	}

frmttable using "TableS16", tex frag replace ///
		statmat(TS16) sdec(3) ///
		annotate(TS16star) asym("*", "**", "***") ///
		cti("" "Partisan attachment" "" "Completed" "Incomplete" "Time to read" "" \ ///
			"Variable" "Government" "Opposition" "secondary (or -)" "college (or +)" "First quintile" "Fifth quintile") ///
		rti("\textit{Argentina}" \ "\hspace{1ex} Optimistic" \ "\hspace{1ex} Joyful" \ "\hspace{1ex} Angry" \ "\hspace{1ex} Sad" \ ///
			"\hspace{1ex} Stressed" \ "\hspace{1ex} Fearful" \ "\hspace{1ex} Disgusted" \ "\hspace{1ex} Indifferent" \ ///
			"\noalign{\smallskip}\textit{Brazil}" \ "\hspace{1ex} Optimistic" \ "\hspace{1ex} Joyful" \ "\hspace{1ex} Angry" \ "\hspace{1ex} Sad" \ ///
			"\hspace{1ex} Stressed" \ "\hspace{1ex} Fearful" \ "\hspace{1ex} Disgusted" \ "\hspace{1ex} Indifferent" \ ///
			"\noalign{\smallskip}\textit{Chile}" \ "\hspace{1ex} Optimistic" \ "\hspace{1ex} Joyful" \ "\hspace{1ex} Angry" \ "\hspace{1ex} Sad" \ ///
			"\hspace{1ex} Stressed" \ "\hspace{1ex} Fearful" \ "\hspace{1ex} Disgusted" \ "\hspace{1ex} Indifferent" \ ///
			"\noalign{\smallskip}\textit{Colombia}" \ "\hspace{1ex} Optimistic" \ "\hspace{1ex} Joyful" \ "\hspace{1ex} Angry" \ "\hspace{1ex} Sad" \ ///
			"\hspace{1ex} Stressed" \ "\hspace{1ex} Fearful" \ "\hspace{1ex} Disgusted" \ "\hspace{1ex} Indifferent") ///
		noblank ///
		hline(101 00 00 00 00 00 00 00 00 01 00 00 00 00 00 00 00 00 01 00 00 00 00 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")



**#*===*      1.2.5. Detailed heterogeneity by partisan attachment (S17-S20)

* Table S17. Difference of means in reactions by partisan attachment

mat TS17 = J(20, 8, .)
mat TS17star = J(20, 8, 0)

local controls = "woman i.age i.education works b1.vote_encode ln_time_read covid i.vax_shots" // income pol_know
local depvars = "engage like share comment"

	local j = 1
	foreach condition in "government == 1" "opposition == 1" "government == 0 & opposition == 0" "DiD opp to gov" {
		
		local i = 2
		foreach cn in "ARG" "BRA" "CHL" "COL" {
				
			foreach v in `depvars' {
				if ("`condition'" != "DiD opp to gov") {
					qui: reg `v' true `controls' if `condition' & country == "`cn'" & treatment != "Dog" & filter == 1, robust
						mat results = r(table)
							mat TS17[`i', `j'] = results["b","true"]
							mat TS17[`i', `j'+1] = results["se","true"]
						local pval = results["pvalue","true"]
							mat TS17star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
				}

				else {
					qui: reg `v' true c.true#b1.vote_encode `controls' if country == "`cn'" & treatment != "Dog" & filter == 1, robust
						mat results = r(table)
							mat TS17[`i',`j'] = results["b","2.vote_encode#c.true"]
							mat TS17[`i',`j'+1] = results["se","2.vote_encode#c.true"]
						local pval = results["pvalue","2.vote_encode#c.true"]
							mat TS17star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
				}
				if (mod(`i',5) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		
		local j = `j' + 2
	}

frmttable using "TableS17", tex frag replace ///
		statmat(TS17) substat(1) sdec(3) ///
		annotate(TS17star) asym("*", "**", "***") ///
		cti("Variable" "Government" "Opposition" "Blank vote" "Opposition v. government") ///
		rti("\textit{Argentina}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Brazil}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Chile}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Colombia}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment") ///
		noblank ///
		hline(11 00 00 00 00 01 00 00 00 00 01 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")


* Table S18. Difference of means in emotions by partisan attachment

mat TS18 = J(36, 8, .)
mat TS18star = J(36, 8, 0)

local controls = "woman i.age i.education works b1.vote_encode ln_time_read covid i.vax_shots" // income pol_know
local depvars = "optimism joy anger sadness stress fear disgust indifference"

	local j = 1
	foreach condition in "government == 1" "opposition == 1" "government == 0 & opposition == 0" "DiD opp to gov" {
		
		local i = 2
		foreach cn in "ARG" "BRA" "CHL" "COL" {
				
			foreach v in `depvars' {
				if ("`condition'" != "DiD opp to gov") {
					qui: reg `v' true `controls' if `condition' & country == "`cn'" & treatment != "Dog" & filter == 1, robust
						mat results = r(table)
							mat TS18[`i', `j'] = results["b","true"]
							mat TS18[`i', `j'+1] = results["se","true"]
						local pval = results["pvalue","true"]
							mat TS18star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
				}

				else {
					qui: reg `v' true c.true#b1.vote_encode `controls' if country == "`cn'" & treatment != "Dog" & filter == 1, robust
						mat results = r(table)
							mat TS18[`i',`j'] = results["b","2.vote_encode#c.true"]
							mat TS18[`i',`j'+1] = results["se","2.vote_encode#c.true"]
						local pval = results["pvalue","2.vote_encode#c.true"]
							mat TS18star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
				}
				if (mod(`i',9) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		
		local j = `j' + 2
	}

frmttable using "TableS18", tex frag replace ///
		statmat(TS18) substat(1) sdec(3) ///
		annotate(TS18star) asym("*", "**", "***") ///
		cti("Variable" "Government" "Opposition" "Blank vote" "Opposition v. government") ///
		rti("\textit{Argentina}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Brazil}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Chile}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Colombia}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent") ///
		noblank ///
		hline(11 00 00 00 00 00 00 00 00 01 00 00 00 00 00 00 00 00 01 00 00 00 00 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")


* Table S19. Levels of engagement by partisan attachment

mat TS19 = J(20, 8, .)

local depvars = "engage like share comment"

	local j = 1
	foreach condition in "government == 1" "opposition == 1" {
		
		local i = 2
		foreach cn in "ARG" "BRA" "CHL" "COL" {
				
			foreach v in `depvars' {
				qui: ttest `v' if `condition' & country == "`cn'" & treatment != "Dog" & filter == 1, by(true)
					mat TS19[`i', `j'] = r(mu_1)
					mat TS19[`i', `j'+1] = (r(sd_1)/sqrt(r(N_1)))
					mat TS19[`i', `j'+2] = r(mu_2)
					mat TS19[`i', `j'+3] = r(sd_2)/sqrt(r(N_2))

				if (mod(`i',5) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		
		local j = `j' + 4
	}

frmttable using "TableS19", tex frag replace ///
		statmat(TS19) substat(1) sdec(3) ///
		cti("" "Government" "" "Opposition" "" \ ///
			"Variable" "Refutation" "Confirmation" "Refutation" "Confirmation") ///
		rti("\textit{Argentina}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Brazil}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Chile}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Colombia}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment") ///
		noblank ///
		hline(101 00 00 00 00 01 00 00 00 00 01 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Standard errors in parentheses.")


* Table S20. Levels of emotions by partisan attachment

mat TS20 = J(36, 8, .)

local depvars = "optimism joy anger sadness stress fear disgust indifference"

	local j = 1
	foreach condition in "government == 1" "opposition == 1" {
		
		local i = 2
		foreach cn in "ARG" "BRA" "CHL" "COL" {
				
			foreach v in `depvars' {
				qui: ttest `v' if `condition' & country == "`cn'" & treatment != "Dog" & filter == 1, by(true)
					mat TS20[`i', `j'] = r(mu_1)
					mat TS20[`i', `j'+1] = (r(sd_1)/sqrt(r(N_1)))
					mat TS20[`i', `j'+2] = r(mu_2)
					mat TS20[`i', `j'+3] = r(sd_2)/sqrt(r(N_2))

				if (mod(`i',9) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		
		local j = `j' + 4
	}

frmttable using "TableS20", tex frag replace ///
		statmat(TS20) substat(1) sdec(3) ///
		cti("" "Government" "" "Opposition" "" \ ///
			"Variable" "Refutation" "Confirmation" "Refutation" "Confirmation") ///
		rti("\textit{Argentina}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Brazil}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Chile}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Colombia}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent") ///
		noblank ///
		hline(101 00 00 00 00 00 00 00 00 01 00 00 00 00 00 00 00 00 01 00 00 00 00 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Standard errors in parentheses.")


		
**#*===*      1.2.6. Detailed heterogeneity by vaccination status (S21-S22)

* Table S21. Effects on reactions by number of shots administered

mat TS21 = J(20, 6, .)
mat TS21star = J(20, 6, 0)

local controls = "woman i.age i.education works i.vote_encode ln_time_read covid" // income pol_know
local depvars = "engage like share comment"

	local j = 1
	foreach condition in "vax_shots == 0" "vax_shots == 1" "vax_shots == 2" {
		
		local i = 2
		foreach cn in "ARG" "BRA" "CHL" "COL" {
				
			foreach v in `depvars' {
				qui: reg `v' true `controls' if `condition' & country == "`cn'" & treatment != "Dog" & filter == 1, robust
					mat results = r(table)
						mat TS21[`i', `j'] = results["b","true"]
						mat TS21[`i', `j'+1] = results["se","true"]
					local pval = results["pvalue","true"]
						mat TS21star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
				
				if (mod(`i',5) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		
		local j = `j' + 2
	}

frmttable using "TableS21", tex frag replace ///
		statmat(TS21) substat(1) sdec(3) ///
		annotate(TS21star) asym("*", "**", "***") ///
		cti("Variable" "Non-vaccinated" "Vaccinated once" "Vaccinated twice+") ///
		rti("\textit{Argentina}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Brazil}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Chile}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Colombia}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment") ///
		noblank ///
		hline(11 00 00 00 00 01 00 00 00 00 01 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")


* Table S22. Effects on emotions by number of shots administered
		
mat TS22 = J(36, 6, .)
mat TS22star = J(36, 6, 0)

local controls = "woman i.age i.education works i.vote_encode ln_time_read covid" // income pol_know
local depvars = "optimism joy anger sadness stress fear disgust indifference"

	local j = 1
	foreach condition in "vax_shots == 0" "vax_shots == 1" "vax_shots == 2" {
		
		local i = 2
		foreach cn in "ARG" "BRA" "CHL" "COL" {
				
			foreach v in `depvars' {
				qui: reg `v' true `controls' if `condition' & country == "`cn'" & treatment != "Dog" & filter == 1, robust
					mat results = r(table)
						mat TS22[`i', `j'] = results["b","true"]
						mat TS22[`i', `j'+1] = results["se","true"]
					local pval = results["pvalue","true"]
						mat TS22star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)

				if (mod(`i',9) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		
		local j = `j' + 2
	}
	
frmttable using "TableS22", tex frag replace ///
		statmat(TS22) substat(1) sdec(3) ///
		annotate(TS22star) asym("*", "**", "***") ///
		cti("Variable" "Non-vaccinated" "Vaccinated once" "Vaccinated twice+") ///
		rti("\textit{Argentina}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Brazil}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Chile}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Colombia}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent") ///
		noblank ///
		hline(11 00 00 00 00 00 00 00 00 01 00 00 00 00 00 00 00 00 01 00 00 00 00 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")
		
		
		
**#*===*      1.2.7. Detailed heterogeneity by education (S23-S24)

* Table S23. Effects on reactions by educational attainment

mat TS23 = J(20, 10, .)
mat TS23star = J(20, 10, 0)

local controls = "woman i.age works i.vote_encode ln_time_read covid i.vax_shots" // income pol_know
local depvars = "engage like share comment"

	local j = 1
	foreach condition in "educ5 == 1" "educ5 == 2" "educ5 == 3" "educ5 == 4" "educ5 == 5" {
		
		local i = 2
		foreach cn in "ARG" "BRA" "CHL" "COL" {
				
			foreach v in `depvars' {
				qui: reg `v' true `controls' if `condition' & country == "`cn'" & treatment != "Dog" & filter == 1, robust
					mat results = r(table)
						mat TS23[`i', `j'] = results["b","true"]
						mat TS23[`i', `j'+1] = results["se","true"]
					local pval = results["pvalue","true"]
						mat TS23star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
						
				if (mod(`i',5) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		
		local j = `j' + 2
	}

frmttable using "TableS23", tex frag replace ///
		statmat(TS23) substat(1) sdec(3) ///
		annotate(TS23star) asym("*", "**", "***") ///
		cti("" "Incomplete" "Completed" "Incomplete" "Completed" "Incomplete" \ ///
			"Variable" "secondary (or -)$^1$" "secondary" "college" "college" "graduate (or +)") ///
		rti("\textit{Argentina}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Brazil}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Chile}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Colombia}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment") ///
		noblank ///
		hline(101 00 00 00 00 01 00 00 00 00 01 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question." "$^1$ This category comprises less than 5% of the observations.")


* Table S24. Effects on emotions by educational attainment
		
mat TS24 = J(36, 10, .)
mat TS24star = J(36, 10, 0)

local controls = "woman i.age works i.vote_encode ln_time_read covid i.vax_shots" // income pol_know
local depvars = "optimism joy anger sadness stress fear disgust indifference"

	local j = 1
	foreach condition in "educ5 == 1" "educ5 == 2" "educ5 == 3" "educ5 == 4" "educ5 == 5" {
		
		local i = 2
		foreach cn in "ARG" "BRA" "CHL" "COL" {
				
			foreach v in `depvars' {
				qui: reg `v' true `controls' if `condition' & country == "`cn'" & treatment != "Dog" & filter == 1, robust
					mat results = r(table)
						mat TS24[`i', `j'] = results["b","true"]
						mat TS24[`i', `j'+1] = results["se","true"]
					local pval = results["pvalue","true"]
						mat TS24star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)

				if (mod(`i',9) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		
		local j = `j' + 2
	}
	
frmttable using "TableS24", tex frag replace ///
		statmat(TS24) substat(1) sdec(3) ///
		annotate(TS24star) asym("*", "**", "***") ///
		cti("" "Incomplete" "Completed" "Incomplete" "Completed" "Incomplete" \ ///
			"Variable" "secondary (or -)$^1$" "secondary" "college" "college" "graduate (or +)") ///
		rti("\textit{Argentina}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Brazil}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Chile}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Colombia}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent") ///
		noblank ///
		hline(101 00 00 00 00 00 00 00 00 01 00 00 00 00 00 00 00 00 01 00 00 00 00 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question." "$^1$ This category comprises less than 5% of the observations.")



**#*===*      1.2.8. Detailed heterogeneity by time to read (S25-S26)

* Table S25. Effects on reactions by quintiles of the time spent reading the question

mat TS25 = J(20, 8, .)
mat TS25star = J(20, 8, 0)

local controls = "woman i.age i.education works i.vote_encode covid i.vax_shots" // income pol_know
local depvars = "engage like share comment"

	local j = 1
	foreach condition in "time_q == 1" "time_q == 3" "time_q == 5" "DiD q5 to q1" {
		
		local i = 2
		foreach cn in "ARG" "BRA" "CHL" "COL" {
				
			foreach v in `depvars' {
				if ("`condition'" != "DiD q5 to q1") {
					qui: reg `v' true `controls' if `condition' & country == "`cn'" & treatment != "Dog" & filter == 1, robust
						mat results = r(table)
							mat TS25[`i', `j'] = results["b","true"]
							mat TS25[`i', `j'+1] = results["se","true"]
						local pval = results["pvalue","true"]
							mat TS25star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
				}

				else {
					qui: reg `v' true c.true#b1.time_q b1.time_q `controls' if country == "`cn'" & treatment != "Dog" & filter == 1, robust
						mat results = r(table)
							mat TS25[`i',`j'] = results["b","5.time_q#c.true"]
							mat TS25[`i',`j'+1] = results["se","5.time_q#c.true"]
						local pval = results["pvalue","5.time_q#c.true"]
							mat TS25star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
				}
				if (mod(`i',5) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		
		local j = `j' + 2
	}

frmttable using "TableS25", tex frag replace ///
		statmat(TS25) substat(1) sdec(3) ///
		annotate(TS25star) asym("*", "**", "***") ///
		cti("" "First" "Third" "Fifth" "Fifth v." \ ///
			"Variable" "quintile" "quintile" "quintile" "first quintile") ///
		rti("\textit{Argentina}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Brazil}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Chile}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Colombia}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment") ///
		noblank ///
		hline(101 00 00 00 00 01 00 00 00 00 01 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")


* Table S26. Effects on emotions by quintiles of the time spent reading the question

mat TS26 = J(36, 8, .)
mat TS26star = J(36, 8, 0)

local controls = "woman i.age i.education works i.vote_encode covid i.vax_shots" // income pol_know
local depvars = "optimism joy anger sadness stress fear disgust indifference"

	local j = 1
	foreach condition in "time_q == 1" "time_q == 3" "time_q == 5" "DiD q5 to q1" {
		
		local i = 2
		foreach cn in "ARG" "BRA" "CHL" "COL" {
				
			foreach v in `depvars' {
				if ("`condition'" != "DiD q5 to q1") {
					qui: reg `v' true `controls' if `condition' & country == "`cn'" & treatment != "Dog" & filter == 1, robust
						mat results = r(table)
							mat TS26[`i', `j'] = results["b","true"]
							mat TS26[`i', `j'+1] = results["se","true"]
						local pval = results["pvalue","true"]
							mat TS26star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
				}

				else {
					qui: reg `v' true c.true#b1.time_q b1.time_q `controls' if country == "`cn'" & treatment != "Dog" & filter == 1, robust
						mat results = r(table)
							mat TS26[`i',`j'] = results["b","5.time_q#c.true"]
							mat TS26[`i',`j'+1] = results["se","5.time_q#c.true"]
						local pval = results["pvalue","5.time_q#c.true"]
							mat TS26star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
				}
				if (mod(`i',9) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		
		local j = `j' + 2
	}

frmttable using "TableS26", tex frag replace ///
		statmat(TS26) substat(1) sdec(3) ///
		annotate(TS26star) asym("*", "**", "***") ///
		cti("" "First" "Third" "Fifth" "Fifth v." \ ///
			"Variable" "quintile" "quintile" "quintile" "first quintile") ///
		rti("\textit{Argentina}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Brazil}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Chile}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Colombia}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent") ///
		noblank ///
		hline(101 00 00 00 00 00 00 00 00 01 00 00 00 00 00 00 00 00 01 00 00 00 00 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")



**#*===*      1.2.9. Detailed heterogeneity by attention filter (S27-S28)

* Table S27. Effects on reactions by attention level

mat TS27 = J(15, 6, .)
mat TS27star = J(15, 6, 0)

local controls = "woman i.age college works i.vote_encode ln_time_read covid i.vax_shots" // income pol_know
local depvars = "engage like share comment"

	local j = 1
	foreach condition in "attention == 0" "attention == 1" "DiD attention" {
		
		local i = 2
		foreach cn in "BRA" "CHL" "COL" {
				
			foreach v in `depvars' {
				if ("`condition'" != "DiD attention") {
					qui: reg `v' true `controls' if `condition' & country == "`cn'" & treatment != "Dog" & filter == 1, robust
						mat results = r(table)
							mat TS27[`i', `j'] = results["b","true"]
							mat TS27[`i', `j'+1] = results["se","true"]
						local pval = results["pvalue","true"]
							mat TS27star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
				}

				else {
					qui: reg `v' true c.true#c.attention attention `controls' if country == "`cn'" & treatment != "Dog" & filter == 1, robust
						mat results = r(table)
							mat TS27[`i',`j'] = results["b","c.attention#c.true"]
							mat TS27[`i',`j'+1] = results["se","c.attention#c.true"]
						local pval = results["pvalue","c.attention#c.true"]
							mat TS27star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
				}
				if (mod(`i',5) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		
		local j = `j' + 2
	}

frmttable using "TableS27", tex frag replace ///
		statmat(TS27) substat(1) sdec(3) ///
		annotate(TS27star) asym("*", "**", "***") ///
		cti("Variable" "Inattentive" "Attentive" "Difference") ///
		rti("\textit{Brazil}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Chile}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Colombia}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment") ///
		noblank ///
		hline(11 00 00 00 00 01 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")


* Table S28. Effects on emotions by attention level
		
mat TS28 = J(27, 6, .)
mat TS28star = J(27, 6, 0)

local controls = "woman i.age i.education works i.vote_encode ln_time_read covid i.vax_shots" // income pol_know
local depvars = "optimism joy anger sadness stress fear disgust indifference"

	local j = 1
	foreach condition in "attention == 0" "attention == 1" "DiD attention" {
		
		local i = 2
		foreach cn in "BRA" "CHL" "COL" {
				
			foreach v in `depvars' {
				if ("`condition'" != "DiD attention") {
					qui: reg `v' true `controls' if `condition' & country == "`cn'" & treatment != "Dog" & filter == 1, robust
						mat results = r(table)
							mat TS28[`i', `j'] = results["b","true"]
							mat TS28[`i', `j'+1] = results["se","true"]
						local pval = results["pvalue","true"]
							mat TS28star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
				}

				else {
					qui: reg `v' true c.true#c.attention attention `controls' if country == "`cn'" & treatment != "Dog" & filter == 1, robust
						mat results = r(table)
							mat TS28[`i',`j'] = results["b","c.attention#c.true"]
							mat TS28[`i',`j'+1] = results["se","c.attention#c.true"]
						local pval = results["pvalue","c.attention#c.true"]
							mat TS28star[`i', `j'] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)
				}
				if (mod(`i',9) == 0) local i = `i' + 2
				else local i = `i' + 1
			}
		}
		
		local j = `j' + 2
	}
	
frmttable using "TableS28", tex frag replace ///
		statmat(TS28) substat(1) sdec(3) ///
		annotate(TS28star) asym("*", "**", "***") ///
		cti("Variable" "Inattentive" "Attentive" "Difference") ///
		rti("\textit{Brazil}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Chile}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent" \ "" \ ///
			"\noalign{\smallskip}\textit{Colombia}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent") ///
		noblank ///
		hline(11 00 00 00 00 00 00 00 00 01 00 00 00 00 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Differences are estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")
		


**#*===*      1.2.10. Placebo experiment by country (S29-S31)

***** Brazil, Chile, and Colombia

local controls = "woman i.age i.education works i.vote_encode ln_time_read covid i.vax_shots" // income pol_know
local depvars = "engage like share comment optimism joy anger sadness stress fear disgust indifference"

local k = 29
foreach cn in "BRA" "CHL" "COL" {
	mat TS`k' = J(14, 8, .)
	mat TS`k'star = J(14, 8, 0)

	local i = 2
	foreach v in `depvars' {

		* Mean values
		qui: ttest `v' if country == "`cn'" & treatment == "Dog" & filter == 1, by(true)
			mat TS`k'[`i', 1] = r(mu_1)
			mat TS`k'[`i', 2] = (r(sd_1)/sqrt(r(N_1)))
			mat TS`k'[`i', 3] = r(mu_2)
			mat TS`k'[`i', 4] = r(sd_2)/sqrt(r(N_2))
		
		* Difference without controls
		qui: reg `v' true if country == "`cn'" & treatment == "Dog" & filter == 1, robust
			mat results = r(table)
				mat TS`k'[`i', 5] = results["b","true"]
				mat TS`k'[`i', 6] = results["se","true"]
			local pval = results["pvalue","true"]
				mat TS`k'star[`i', 5] = (`pval' <= .05) + (`pval' <= .01) + (`pval' <= .001)

		* Demographic and political controls + time to read
		qui: reg `v' true `controls' if country == "`cn'" & treatment == "Dog" & filter == 1, robust
			mat results = r(table)
				mat TS`k'[`i',7] = results["b","true"]
				mat TS`k'[`i',8] = results["se","true"]
			local pval = results["pvalue","true"]
				mat TS`k'star[`i',7] = (`pval' < .05) + (`pval' < .01) + (`pval' < .001)

		if (`i' == 5) local i = `i' + 2
		else local i = `i' + 1
	}

	frmttable using "TableS`k'", tex frag replace ///
		statmat(TS`k') substat(1) sdec(3) ///
		annotate(TS`k'star) asym("*", "**", "***") ///
		cti("" "Refutation" "Confirmation" "Simple" "Difference" \ ///
			"Variable" "frame" "frame" "difference" "with controls") ///
		rti("\textit{Reactions}" \ "" \ "\hspace{1ex} Engage" \ "" \ "\hspace{1ex} Like" \ "" \ "\hspace{1ex} Share" \ "" \ "\hspace{1ex} Comment" \ "" \ ///
			"\noalign{\smallskip}\textit{Emotions}" \ "" \ "\hspace{1ex} Optimistic" \ "" \ "\hspace{1ex} Joyful" \ "" \ "\hspace{1ex} Angry" \ "" \ "\hspace{1ex} Sad" \ "" \ ///
			"\hspace{1ex} Stressed" \ "" \ "\hspace{1ex} Fearful" \ "" \ "\hspace{1ex} Disgusted" \ "" \ "\hspace{1ex} Indifferent") ///
		noblank ///
		hline(101 00 00 00 00 01 {0} 1) ///
		note("\smallskip\textit{Note}: Robust standard errors in parentheses." "*** p $<$ 0.001, ** p $<$ 0.01, * p $<$ 0.05." "Significance levels are derived from linear regression models." "The second difference is estimated controlling for gender, age, educational" "achievement, employment status, partisan" "attachment, and time spent reading the question.")

	local k = `k' + 1
}



****===========================================================================*
**#*===* 2 Figures

graph set window fontface "Helvetica"
cap n mkdir "$figures/$today"
	cd "$figures/$today"

**#*===*   2.1. Figures in main article

global color1 "228 92 92"
global color2 "92 161 92"
global color3 "189 222 189"
global color4 "255 189 189"


**#*===*      2.1.1. Framing effects on reactions and emotions by country

mat def Fig = J(48, 6, .)

local row = 1
local country = 1
foreach c in ARG BRA CHL COL {
	local variable = 2
	foreach v in engage like share comment optimism joy anger sadness stress fear disgust indifference {
		qui: reg `v' true if country == "`c'" & treatment != "Dog" & filter == 1, r
			mat Fig[`row',1] = `country'
			mat Fig[`row',2] = `variable'
			mat Fig[`row',3] = _b[_cons]
			mat Fig[`row',4] = _b[_cons] + _b[true]
			mat Fig[`row',5] = _b[true]
			mat Fig[`row',6] = r(table)["pvalue", "true"]
		
		if (`variable' != 8) local variable = `variable' + 2
			else local variable = `variable' + 4
		local row = `row' + 1
	}
	local country = `country' + 1
}

preserve
	clear
	svmat Fig

	gen labels = strofreal(Fig5, "%9.4f")
		replace labels = labels + "*" if Fig6 < .05
		replace labels = labels + "*" if Fig6 < .01
		replace labels = labels + "*" if Fig6 < .001
		
	rename Fig1 country
	rename Fig2 posall
		gen posall0 = posall - .4
		gen posall1 = posall + .4
	rename Fig3 value0
	rename Fig4 value1
	rename Fig5 diff
	
	la def country 1 "Argentina" 2 "Brazil" 3 "Chile" 4 "Colombia"
		la val country country
	la def variables 2 "Engage" 4 "Like" 6 "Share" 8 "Comment" 12 "Optimistic" 14 "Joyful" 16 "Angry" 18 "Sad" 20 "Stressed" 22 "Fearful" 24 "Disgusted" 26 "Indifferent"
		la val posall posall0 posall1 variables
	
	twoway (bar value0 posall0 if posall < 10, barw(.3) col("$color1")) (bar value1 posall1 if posall < 10, barw(.3) col("$color2")) ///
			(rbar value1 value0 posall if diff > 0 & posall < 10, barw(.3) col("$color3")) ///
				(sc value1 posall if diff > 0 & posall < 10, m(none) mlab(labels) mlabcol(black*.9) mlabpos(12) mlabsi(vsmall)) ///
			(rbar value1 value0 posall if diff < 0 & posall < 10, barw(.3) col("$color4")) ///
				(sc value0 posall if diff < 0 & posall < 10, m(none) mlab(labels) mlabcol(black*.9) mlabpos(12) mlabsi(vsmall)) ///
			, by(country, col(2) note("")) ///
			  xlab(2 4 6 8, value labsize(vsmall)) $gph_by_op ///
			  xsc(range(1.2 8.8)) ///
			  subti(, pos(12) size(small)) ///
			  legend(order(1 "Refutation frame" 2 "Confirmation frame" 3 "Positive difference" 5 "Negative difference") size(vsmall)) ///
			  ylab(, labsize(small)) ///
			  yti("Share of respondents", size(small)) ///
			  xsize(7) ysize(4) name(reactions, replace)
		graph export "Effects_reactions.png", width(4000) replace
		graph export "Effects_reactions.eps", replace
		
	twoway (bar value0 posall0 if posall > 10, barw(.3) col("$color1")) (bar value1 posall1 if posall > 10, barw(.3) col("$color2")) ///
			(rbar value1 value0 posall if diff > 0 & posall > 10, barw(.3) col("$color3")) ///
				(sc value1 posall if diff > 0 & posall > 10, m(none) mlab(labels) mlabcol(black*.9) mlabpos(12) mlabsi(vsmall)) ///
			(rbar value1 value0 posall if diff < 0 & posall > 10, barw(.3) col("$color4")) ///
				(sc value0 posall if diff < 0 & posall > 10, m(none) mlab(labels) mlabcol(black*.9) mlabpos(12) mlabsi(vsmall)) ///
			, by(country, col(2) note("")) ///
			  xlab(12 14 16 18 20 22 24 26, value labsize(vsmall)) $gph_by_op ///
			  xsc(range(11.2 26.8)) ///
			  subti(, pos(12) size(small)) ///
			  legend(order(1 "Refutation frame" 2 "Confirmation frame" 3 "Positive difference" 5 "Negative difference") size(vsmall)) ///
			  ylab(, labsize(small)) ///
			  yti("Share of respondents", size(small)) ///
			  xsize(7) ysize(4) name(emotions, replace)
		graph export "Effects_emotions.png", width(4000) replace
		graph export "Effects_emotions.eps", replace

	twoway (bar value0 posall0, barw(.3) col("$color1")) (bar value1 posall1, barw(.3) col("$color2")) ///
			(rbar value1 value0 posall if diff > 0, barw(.3) col("$color3")) ///
				(sc value1 posall if diff > 0, m(none) mlab(labels) mlabcol(black*.9) mlabpos(12) mlabsi(tiny)) ///
			(rbar value1 value0 posall if diff < 0, barw(.3) col("$color4")) ///
				(sc value0 posall if diff < 0, m(none) mlab(labels) mlabcol(black*.9) mlabpos(12) mlabsi(tiny)) ///
			, by(country, col(1) note("")) ///
			  xlab(2 4 6 8 12 14 16 18 20 22 24 26, value labsize(vsmall) alt) $gph_by_op ///
			  xsc(range(1.2 26.8)) ///
			  subti(, pos(12) size(vsmall)) legend(order(1 "Refutation frame" 2 "Confirmation frame" 3 "Positive difference" 5 "Negative difference") size(vsmall)) ///
			  ylab(, labsize(vsmall)) ///
			  yti("Share of respondents", size(vsmall)) ///
			  xsize(50) ysize(60) name(all, replace)
		graph export "Effects_all.png", width(4000) replace
		graph export "Effects_all.eps", replace
restore


**#*===*      2.1.2. Engagement rate by vaccine brand and frame in Argentina

mat def Fig = J(3, 6, .)

local row = 1
local vaccine = 2
foreach t in AstraZeneca Sputnik Moderna {
	qui: reg engage true if country == "ARG" & treatment == "`t'" & filter == 1, r
		mat Fig[`row',1] = 1
		mat Fig[`row',2] = `vaccine'
		mat Fig[`row',3] = _b[_cons]
		mat Fig[`row',4] = _b[_cons] + _b[true]
		mat Fig[`row',5] = _b[true]
		mat Fig[`row',6] = r(table)["pvalue", "true"]
	
	local vaccine = `vaccine' + 2
	local row = `row' + 1
}

preserve
	clear
	svmat Fig

	gen labels = strofreal(Fig5, "%9.4f")
		replace labels = labels + "*" if Fig6 < .05
		replace labels = labels + "*" if Fig6 < .01
		replace labels = labels + "*" if Fig6 < .001
		
	rename Fig1 country
	rename Fig2 posall
		gen posall0 = posall - .4
		gen posall1 = posall + .4
	rename Fig3 value0
	rename Fig4 value1
	rename Fig5 diff
	
	la def country 1 "Argentina" 2 "Brazil" 3 "Chile" 4 "Colombia"
		la val country country
	la def vaccine 2 "AstraZeneca" 4 "Sputnik V" 6 "Moderna"
		la val posall posall0 posall1 vaccine
	
	twoway (bar value0 posall0 if posall < 10, barw(.3) col("$color1")) (bar value1 posall1 if posall < 10, barw(.3) col("$color2")) ///
			(rbar value1 value0 posall if diff > 0 & posall, barw(.3) col("$color3")) ///
				(sc value1 posall if diff > 0 & posall, m(none) mlab(labels) mlabcol(black*.9) mlabpos(12) mlabsi(small)) ///
			(rbar value1 value0 posall if diff < 0 & posall, barw(.3) col("$color4")) ///
				(sc value0 posall if diff < 0 & posall, m(none) mlab(labels) mlabcol(black*.9) mlabpos(12) mlabsi(small)) ///
			, xlab(2 4 6, value labsize(medsmall)) $gph_gral_op ///
			xsc(range(1.2 6.8)) ///
			legend(order(1 "Refutation frame" 2 "Confirmation frame" 3 "Positive difference" 5 "Negative difference") size(small) pos(6)) ///
			ylab(0(.1).4, labsize(medsmall)) yti("Engagement rate", size(medsmall)) ///
			xsize(7) ysize(4) name(vaccines, replace)
		graph export "Effects_engage_vaccine.png", width(4000) replace
		graph export "Effects_engage_vaccine.eps", replace
restore


**#*===*      2.1.3. Engagement rate by use of labels in Brazil, Chile, and Colombia

mat def Fig = J(6, 6, .)

local row = 1
local country = 2
foreach c in BRA CHL COL {
	local treatment = 2
	foreach t in "Label" "No-label" {
		qui: reg engage true if country == "`c'" & treatment == "`t'" & filter == 1, r
			mat Fig[`row',1] = `country'
			mat Fig[`row',2] = `treatment'
			mat Fig[`row',3] = _b[_cons]
			mat Fig[`row',4] = _b[_cons] + _b[true]
			mat Fig[`row',5] = _b[true]
			mat Fig[`row',6] = r(table)["pvalue", "true"]
		
		local treatment = `treatment' + 2
		local row = `row' + 1
	}
	local country = `country' + 1
}

preserve
	clear
	svmat Fig

	gen labels = strofreal(Fig5, "%9.4f")
		replace labels = labels + "*" if Fig6 < .05
		replace labels = labels + "*" if Fig6 < .01
		replace labels = labels + "*" if Fig6 < .001
		
	rename Fig1 country
	rename Fig2 posall
		gen posall0 = posall - .4
		gen posall1 = posall + .4
	rename Fig3 value0
	rename Fig4 value1
	rename Fig5 diff
	
	la def country 1 "Argentina" 2 "Brazil" 3 "Chile" 4 "Colombia"
		la val country country
	la def treatment 2 "With label" 4 "Without label"
		la val posall posall0 posall1 treatment
	
	twoway (bar value0 posall0 if posall < 10, barw(.3) col("$color1")) (bar value1 posall1 if posall < 10, barw(.3) col("$color2")) ///
			(rbar value1 value0 posall if diff > 0 & posall, barw(.3) col("$color3")) ///
				(sc value1 posall if diff > 0 & posall, m(none) mlab(labels) mlabcol(black*.9) mlabpos(12) mlabsi(small)) ///
			(rbar value1 value0 posall if diff < 0 & posall, barw(.3) col("$color4")) ///
				(sc value0 posall if diff < 0 & posall, m(none) mlab(labels) mlabcol(black*.9) mlabpos(12) mlabsi(small)) ///
			, by(country, col(3) note("")) ///
			xlab(2 4, value labsize(medsmall)) ///
			xsc(range(1.2 4.8)) $gph_by_op ///
			legend(order(1 "Refutation frame" 2 "Confirmation frame" 3 "Positive difference" 5 "Negative difference") size(small) pos(6)) ///
			subti(, pos(12) size(medsmall)) ///
			ylab(0(.1).6, labsize(medsmall)) yti("Engagement rate", size(medsmall)) ///
			xsize(7) ysize(4) name(labels, replace)
		graph export "Effects_engage_labels.png", width(4000) replace
		graph export "Effects_engage_labels.eps", replace
restore


**#*===*      2.1.4. Engagement rate in dog placebo by frame in Brazil, Chile, and Colombia

mat def Fig = J(6, 6, .)

local row = 1
local country = 2
foreach c in BRA CHL COL {
	local treatment = 2
	foreach t in `"treatment != "Dog""' `"treatment == "Dog""' {
		qui: reg engage true if country == "`c'" & filter == 1 & `t', r
			mat Fig[`row',1] = `country'
			mat Fig[`row',2] = `treatment'
			mat Fig[`row',3] = _b[_cons]
			mat Fig[`row',4] = _b[_cons] + _b[true]
			mat Fig[`row',5] = _b[true]
			mat Fig[`row',6] = r(table)["pvalue", "true"]
		
		local treatment = `treatment' + 2
		local row = `row' + 1
	}
	local country = `country' + 1
}

preserve
	clear
	svmat Fig

	gen labels = strofreal(Fig5, "%9.4f")
		replace labels = labels + "*" if Fig6 < .05
		replace labels = labels + "*" if Fig6 < .01
		replace labels = labels + "*" if Fig6 < .001
		
	rename Fig1 country
	rename Fig2 posall
		gen posall0 = posall - .4
		gen posall1 = posall + .4
	rename Fig3 value0
	rename Fig4 value1
	rename Fig5 diff
	
	la def country 1 "Argentina" 2 "Brazil" 3 "Chile" 4 "Colombia"
		la val country country
	la def treatment 2 "Vaccines treatment" 4 "Dog treatment"
		la val posall posall0 posall1 treatment
	
	twoway (bar value0 posall0 if posall < 10, barw(.3) col("$color1")) (bar value1 posall1 if posall < 10, barw(.3) col("$color2")) ///
			(rbar value1 value0 posall if diff > 0 & posall, barw(.3) col("$color3")) ///
				(sc value1 posall if diff > 0 & posall, m(none) mlab(labels) mlabcol(black*.9) mlabpos(12) mlabsi(small)) ///
			(rbar value1 value0 posall if diff < 0 & posall, barw(.3) col("$color4")) ///
				(sc value0 posall if diff < 0 & posall, m(none) mlab(labels) mlabcol(black*.9) mlabpos(12) mlabsi(small)) ///
			, by(country, col(3) note("")) ///
			xlab(2 4, value labsize(medsmall)) ///
			xsc(range(1.2 4.8)) $gph_by_op ///
			legend(order(1 "Refutation frame" 2 "Confirmation frame" 3 "Positive difference" 5 "Negative difference") size(small) pos(6)) ///
			subti(, pos(12) size(medsmall)) ///
			ylab(0(.1).6, labsize(medsmall)) yti("Engagement rate", size(medsmall)) ///
			xsize(7) ysize(4) name(placebo, replace)
		graph export "Effects_engage_placebo.png", width(4000) replace
		graph export "Effects_engage_placebo.eps", replace
restore


**#*===*      2.1.5. Probability of engagement by time to read

reg engage c.ln_time_read##true woman i.age i.education works i.vote_encode covid i.vax_shots if filter == 1 & treatment != "Dog", r
margins, by(true) at(ln_time_read=(1(.2)5))
marginsplot, recast(line) ///
		plot1op(col("216 18 18")) plot2op(col("18 118 18")) ///
		ciop(recast(rarea)) ci1op(col("252 235 235")) ci2op(col("235 244 235")) ///
		$gph_gral_op ///
		ylab(, labsize(medsmall)) xlab(1/5, labsize(medsmall)) ///
		ti("") yti("Predicted probability of 'Engage'", size(medsmall)) xti("Time-to-Read (Log)", margin(small) size(medsmall)) ///
		legend(order(3 "Refutation frame" 4 "Confirmation frame") pos(6) size(small))
	graph export "Pr_engage_by_time.png", width(4000) replace
	graph export "Pr_engage_by_time.eps", replace


**#*===*   2.2. Figures in SIF

**#*===*      2.2.1. Engagement rate by partisan attachment and frame in Argentina

mat def Fig = J(3, 6, .)

local row = 1
local party = 2
foreach t in FdT Cambiemos Blanco {
	qui: reg engage true if country == "ARG" & vote_party == "`t'" & filter == 1, r
		mat Fig[`row',1] = 1
		mat Fig[`row',2] = `party'
		mat Fig[`row',3] = _b[_cons]
		mat Fig[`row',4] = _b[_cons] + _b[true]
		mat Fig[`row',5] = _b[true]
		mat Fig[`row',6] = r(table)["pvalue", "true"]
	
	local party = `party' + 2
	local row = `row' + 1
}

preserve
	clear
	svmat Fig

	gen labels = strofreal(Fig5, "%9.4f")
		replace labels = labels + "*" if Fig6 < .05
		replace labels = labels + "*" if Fig6 < .01
		replace labels = labels + "*" if Fig6 < .001
		
	rename Fig1 country
	rename Fig2 posall
		gen posall0 = posall - .4
		gen posall1 = posall + .4
	rename Fig3 value0
	rename Fig4 value1
	rename Fig5 diff
	
	la def country 1 "Argentina" 2 "Brazil" 3 "Chile" 4 "Colombia"
		la val country country
	la def party 2 "Frente de Todos" 4 "Cambiemos" 6 "Blanco"
		la val posall posall0 posall1 party
	
	twoway (bar value0 posall0 if posall < 10, barw(.3) col("$color1")) (bar value1 posall1 if posall < 10, barw(.3) col("$color2")) ///
			(rbar value1 value0 posall if diff > 0 & posall, barw(.3) col("$color3")) ///
				(sc value1 posall if diff > 0 & posall, m(none) mlab(labels) mlabcol(black*.9) mlabpos(12) mlabsi(small)) ///
			(rbar value1 value0 posall if diff < 0 & posall, barw(.3) col("$color4")) ///
				(sc value0 posall if diff < 0 & posall, m(none) mlab(labels) mlabcol(black*.9) mlabpos(12) mlabsi(small)) ///
			, xlab(2 4 6, value labsize(medsmall)) $gph_gral_op ///
			xsc(range(1.2 6.8)) ///
			legend(order(1 "Refutation frame" 2 "Confirmation frame" 3 "Positive difference" 5 "Negative difference") size(small) pos(6)) ///
			ylab(0(.1).4, labsize(medsmall)) yti("Engagement rate", size(medsmall)) ///
			xsize(7) ysize(4) name(vaccines, replace)
		graph export "Effects_engage_party.png", width(4000) replace
		graph export "Effects_engage_party.eps", replace
restore
